

Calendar Conversion Program Used to Analyze Early History of Islam

Thomas Djamaluddin

Center for Application of Space Science
National Institute of Aeronautics and Space (LAPAN)
Bandung, Indonesia
t_djamil@bdg.lapan.go.id; t_djamil@hotmail.com

Abstract

A simple program to convert Hijri (Islamic) calendar to Julian/Gregorian Calendar has been made. This can be used to analyze early history of Islam, to verify the date, the day, or the season any event during the life of Prophet Muhammad PBUH. Some important events can be concluded as follows. The date of descending of the Quran either on 17 Ramadhan 13 BH (before Hijrah) (Monday 25 August 609 AD) or 12 Rabiul Awal 13 BH (Monday 24 February 609 AD). It is confirmed that the day of hijrah was 12 Rabiul Awal 0 AH on Monday 5 October 621 AD. The day of Badar war 17 Ramadhan 2 AH was not Friday as mentioned in history book, but on Tuesday (13 March 624). There was 9 times fasting of Ramadhan during the life of the Prophet, mostly (6 times) 29 days per month. This confirms hadith from Ibnu Mas'ud and Aisha that the Prophet fasted in Ramadhan mostly 29 days rather than 30 days. The Prophet passed away on 12 Rabiul Awal 11 AH. It was not Monday as written in history book, but Saturday 6 June 632 AD. Was there two days delay of crescent observation due to public sadness during sickness of the prophet that causing Monday 8 June 632 considered as 12 Rabiul Awal?

1. INTRODUCTION

Islamic calendar (Hijriyyah) is determined based on observable new moon (hilaal) with mean period of 29.5306 days. So that one Islamic lunar calendar is 354.3671 days. It is known that 1 Muharram 1 AH was 16 July 622 AD, which can be rechecked by using the program. Then the date in Hijriyyah can be calculated by using those average monthly and yearly day number.

Meanwhile solar calendar (Gregorian/Julian) is determined based on the mean period of position of the sun passing vernal equinox point, i. e. 365.2422 days. The day number in any month is determined conventionally close to lunation, but with additional day to fit yearly day number. January to be 31 days, February 28 or 29 days, March 31 days, April 30 days and so on. Before Gregorian reformation in 1582, Julian calendar use 365.25 per year, 365 days for three common years and 366 days for leap years. Gregorian calendar consist of 365.2425 days per year, close to the real day number. Consequently, there was a date gap between Thursday 4 October and Friday 15 October 1582, without changing the day sequence.

There is the year zero in Islamic calendar (although at that time the concept of number zero was still unknown), that was when the Prophet did hijrah (migration). This is different from Christian solar calendar in which the year started from 1 that made public confusion concerning the beginning of third millenium (1 January 2000 or 2001). It is wrong to mention that hijrah done by the Prophet in September 622 AD, as written in some history books or in sites of online calendar conversion (e.g. [//www.cs.pitt.edu/~tawfig/convert/introduction.html](http://www.cs.pitt.edu/~tawfig/convert/introduction.html) and [//bennyhills.fortunecity.com/elfman/454/calindex.html#about](http://bennyhills.fortunecity.com/elfman/454/calindex.html#about) which refers to

intro.html). Hijrah took place on 5 October 621, i. e. 12 Rabiul Awal 0, which is consistent with the next New Year 1 Muharram 1 AH on 16 July 622 AD.

When I was taking my graduate school (1991) in Department of Astronomy, Kyoto University, Japan, I was asked by a professor of South-East Asian Studies to develop a computer program to convert any date in Islamic lunar calendar to solar calendar, and vice versa. This simple program can be used in analyzing any Islamic historical data, including the early history of Islam during the life of the Prophet Muhammad PBUH. This paper reports some analysis of the early history of Islam, by comparing day and season written in history books with results of astronomical approach.

2. THE PROGRAM

In principle, the program can be used to convert calendar since 1 January 1 AD to 31 December 2099. However, the user must be aware that the output may be different from the actual date (plus or minus one day), since the program calculation use only mean period of lunation. It is more difficult to make a conversion program with the output to be local actual date. The actual hilaal sighting is affected by atmospheric condition that cannot be formulated for long term calendar conversion program. Due to such uncertainty, in the program there is no information what day is a lunar calendar date. The day can be determined only for conversion from solar calendar to lunar calendar.

The inputs of the program are

- a. Conversion type:
 1. CONVERSION HIJRI(ISLAMIC CALENDAR) TO SOLAR CALENDAR.
 2. CONVERSION SOLAR CALENDAR TO HIJRI(ISLAMIC CALENDAR).
- b. Date, month, and year.

Examples of the result are shown in the Table 1. It is possible to input year as zero (for calculating the day of hijrah from Mecca to Madina) or minus (for calculating the birthday of the Prophet and other events before Hijrah).

Table 1. Program's Results

Conversion type	Input	Output
1	1 Muharram 1 AH	16 July 622 AD
2	16 July 622 AD	1 Muharram 1 AH (Friday)
1	16 Ramadhan 990 AH	4 October 1582 AD
2	4 October 1582 AD	16 Ramadhan 990 H (Thursday)
1	17 Ramadhan 990 AH	15 October 1582 AD
2	15 October 1582 AD	17 Ramadhan 990 H (Friday)
1	8 Ramadhan 1364 AH	*18 August 1945 AD
2	17 August 1945 AD *	8 Ramadhan 1364 (Friday)
1	1 Shawal 1422 AH*	17 December 2001
2	17 December 2001 AD	*30 Ramadhan 1422 AH

* : "Different date" as explained in the text.

```

10 PRINT "*****CALENDAR CONVERSION PROGRAM *****"
20 PRINT " HIJRI(ISLAMIC CALENDAR) <--> SOLAR CALENDAR "
30 PRINT " (CAUTION: THERE IS THE ISLAMIC DATE LINE  "
40 PRINT "      WHICH CAUSES THIS CALCULATION MAY "
50 PRINT "      DIFFER ABOUT (+/-) 1 DAY)  "
60 PRINT " [6 AUGUST 1991, 25 MUHARRAM 1412]  "

```

```

70 PRINT "    CALCULATED BY T.DJAMALUDDIN    "
80 PRINT "  DEPT. OF ASTRONOMY, KYOTO UNIVERSITY  "
90 PRINT "    SAKYO-KU, KYOTO 606, JAPAN    "
100 PRINT "*****"
110 PRINT
120 PRINT "1. CONVERSION HIJRI(ISLAMIC CALENDAR) TO SOLAR CALENDAR"
130 PRINT "2. CONVERSION SOLAR CALENDAR TO HIJRI(ISLAMIC CALENDAR)"
140 PRINT
150 INPUT "CONVERSION (1 OR 2) OR STOP(0)"; CONV
160 IF CONV=2 GOTO 820
165 IF CONV=0 THEN GOTO 1500
170 PRINT "***CONVERSION HIJRI TO SOLAR CALENDAR***"
180 PRINT
190 INPUT " DATE                ";HD
200 PRINT " 1. MUHARRAM      7. RAJAB  "
210 PRINT " 2. SAFAR         8. SHABAN  "
220 PRINT " 3. RABIUL AWAL   9. RAMADHAN  "
230 PRINT " 4. RABIUTH THANI 10. SHAWAL  "
240 PRINT " 5. JUMADIL ULA  11. ZULQAIDAH  "
250 PRINT " 6. JUMADITH THANI 12. ZULHIJAH  "
260 INPUT " MONTH                ";HM
270 INPUT " YEAR                    ";HY
280 NHD1=(HY-1)*354.3671+(HM-1)*29.5306+HD : NHD = INT(NHD1)
290 NSD = NHD1 + 227016!
300 IF NHD > 350721! THEN GC = 10 ELSE GC = 0
310 IF NHD > 393898! THEN GC = 11
320 IF NHD > 430422! THEN GC = 12
330 IF NHD > 466946! THEN GC = 13
340 SY = INT((NSD+GC)/365.25) + 1
350 MN = CINT((NSD+GC)-(SY-1)*365.25)
360 MN1 = 0 : SM = 1
370 IF MN>31 THEN MN1 = 31 : SM = 2
380 IF INT(SY/4) = SY/4 THEN GOSUB 700 ELSE GOSUB 580
390 IF SY = 1700 OR SY = 1800 THEN GOSUB 580
400 IF SY = 1900 THEN GOSUB 580
410 *****RESULT*****
420 IF SM = 1 THEN SM$ = " JANUARY "
430 IF SM = 2 THEN SM$ = " FEBRUARY "
440 IF SM = 3 THEN SM$ = " MARCH "
450 IF SM = 4 THEN SM$ = " APRIL "
460 IF SM = 5 THEN SM$ = " MAY "
470 IF SM = 6 THEN SM$ = " JUNE "
480 IF SM = 7 THEN SM$ = " JULY "
490 IF SM = 8 THEN SM$ = " AUGUST "
500 IF SM = 9 THEN SM$ = " SEPTEMBER "
510 IF SM =10 THEN SM$ = " OCTOBER "
520 IF SM =11 THEN SM$ = " NOVEMBER "
530 IF SM =12 THEN SM$ = " DECEMBER "
540 SD = MN - MN1
550 PRINT
560 PRINT HD;" ";HM;" ";HY;" HIJRI IS ";SD;SM$;SY;"A.D."
570 PRINT : PRINT: GOTO 110
580 *****ORDINARY (BASITAH) YEAR*****
590 IF MN > 59 THEN MN1 = 59 : SM = 3
600 IF MN > 90 THEN MN1 = 90 : SM = 4
610 IF MN >120 THEN MN1 =120 : SM = 5
620 IF MN >151 THEN MN1 =151 : SM = 6
630 IF MN >181 THEN MN1 =181 : SM = 7
640 IF MN >212 THEN MN1 =212 : SM = 8
650 IF MN >243 THEN MN1 =243 : SM = 9

```

```

660 IF MN >273 THEN MN1 =273 : SM =10
670 IF MN >304 THEN MN1 =304 : SM =11
680 IF MN >334 THEN MN1 =334 : SM =12
690 RETURN
700 *****LEAP (KABISAT) YEAR*****
710 IF MN > 60 THEN MN1 = 60 : SM = 3
720 IF MN > 91 THEN MN1 = 91 : SM = 4
730 IF MN >121 THEN MN1 =121 : SM = 5
740 IF MN >152 THEN MN1 =152 : SM = 6
750 IF MN >182 THEN MN1 =182 : SM = 7
760 IF MN >213 THEN MN1 =213 : SM = 8
770 IF MN >244 THEN MN1 =244 : SM = 9
780 IF MN >274 THEN MN1 =274 : SM =10
790 IF MN >305 THEN MN1 =305 : SM =11
800 IF MN >335 THEN MN1 =335 : SM =12
810 RETURN
820 PRINT "***SOLAR CALENDAR TO HIJRI***"
830 PRINT
840 INPUT " DATE           ";SD
850 PRINT " 1. JANUARY       7. JULY      "
860 PRINT " 2. FEBRUARY       8. AUGUST   "
870 PRINT " 3. MARCH          9. SEPTEMBER "
880 PRINT " 4. APRIL          10. OCTOBER  "
890 PRINT " 5. MAY            11. NOVEMBER "
900 PRINT " 6. JUNE           12. DECEMBER "
910 INPUT " MONTH           ";SM
920 INPUT " YEAR            ";SY
930 IF SM = 1 THEN MN = 0
940 IF SM = 2 THEN MN = 31
950 IF SM = 3 THEN MN = 59
960 IF SM = 4 THEN MN = 90
970 IF SM = 5 THEN MN =120
980 IF SM = 6 THEN MN =151
990 IF SM = 7 THEN MN =181
1000 IF SM = 8 THEN MN =212
1010 IF SM = 9 THEN MN =243
1020 IF SM =10 THEN MN =273
1030 IF SM =11 THEN MN =304
1040 IF SM =12 THEN MN =334
1050 Y = SY + (MN+SD)/365.25
1060 IF Y>1582.76 AND Y<1582.788 THEN PRINT "GREGORIAN : OMITTED": GOTO 110
1070 IF Y>1582.76 THEN GC = 10 ELSE GC = 0
1080 IF Y>1701 THEN GC = 11
1090 IF Y>1801 THEN GC = 12
1100 IF Y>1901 THEN GC = 13
1110 YDN = (SY-1)*365.25
1120 IF SM < 3 THEN GOTO 1160
1130 IF INT(SY/4) = SY/4 THEN MN = MN+1
1140 IF SY = 1700 THEN MN = MN - 1
1150 IF SY = 1800 OR SY = 1900 THEN MN = MN - 1
1160 NSD = YDN + MN + SD - GC
1170 NHD1= NSD - 227016! : NHD = INT(NHD1)
1180 HY = INT(NHD1/354.3671 + 1)
1190 HM1 = CINT(NHD1 - (HY-1)*354.3671)
1200 HM = INT(HM1/29.5306) + 1
1210 HD = CINT(HM1 - (HM-1)*29.5306 ) : GOSUB 1400
1220 IF HD = 0 THEN HM = HM-1 : HD = 30
1230 *****RESULT*****
1240 IF HM = 0 THEN HM$ = " ZULHIJAH " : HY = HY - 1
1250 IF HM = 1 THEN HM$ = " MUHARRAM "

```

```

1260 IF HM = 2 THEN HM$ = " SAFAR "
1270 IF HM = 3 THEN HM$ = " RABIUL AWAL "
1280 IF HM = 4 THEN HM$ = " RABIUTH THANI "
1290 IF HM = 5 THEN HM$ = " JUMADIL ULA "
1300 IF HM = 6 THEN HM$ = " JUMADITH THANI"
1310 IF HM = 7 THEN HM$ = " RAJAB "
1320 IF HM = 8 THEN HM$ = " SHABAN "
1330 IF HM = 9 THEN HM$ = " RAMADHAN "
1340 IF HM =10 THEN HM$ = " SHAWAL "
1350 IF HM =11 THEN HM$ = " ZULQAIDAH "
1360 IF HM =12 THEN HM$ = " ZULHIJAH "
1370 PRINT
1380 PRINT SD;".";SM;".";SY;" A.D IS "; HD; HM$; HY; " HIJRI ";"(;"DD$;")"
1390 PRINT : PRINT: GOTO 110
1400 '*****THE DAY*****
1410 NHD1 = NHD-7*INT(NHD/7) : DD = NHD1 MOD 7
1420 IF DD = 0 THEN DD$ = " THURSDAY "
1430 IF DD = 1 THEN DD$ = " FRIDAY "
1440 IF DD = 2 THEN DD$ = " SATURDAY "
1450 IF DD = 3 THEN DD$ = " SUNDAY "
1460 IF DD = 4 THEN DD$ = " MONDAY "
1470 IF DD = 5 THEN DD$ = " TUESDAY "
1480 IF DD = 6 THEN DD$ = " WEDNESDAY "
1490 RETURN
1500 SYSTEM

```

3. ANALYSIS OF EARLY HISTORY OF ISLAM

In a hadith concerning the fasting on Monday, it is mentioned that the Prophet PBUH said that, that day (Monday) he was born, he became a prophet, and the day of the Quran revealed for the first time (Hadith narrated by Muslim). While Jabir and Ibnu Abbas thought that the Prophet was born in the night of Monday 12 Rabiul Awal, and on the same day and date he became a messenger and a prophet, did *miraj* to the heaven, did hijrah to Madina, and passed away (Al Husaini, 1996).

Various information were found in Islamic history books concerning the events. Haekal (1993), for example, mention that there are many opinions about the births day of the Prophet. The dates are 2, 8, 9, or 12. While the months are Muharram, Safar, Rabiul Awal, Rajab, or Ramadhan. And the years are in "the year of elephant", 15 years before, 30 years after, or 70 years after. The majority, however, thought that the Prophet was born on Monday 12 Rabiul Awal in "the year of elephant". "The year of elephant" was that when Abraha and his army by using elephant intent to destroy the Kaba, but failed. This event occurred 53 years before hijrah (it can be written astronomical-mathematically as -53 H). Calculated by using the program and considering probable error of one day, the birth day of the Prophets was on Monday 5 May 570 AD.

Majority of scholars' opinion say that he became a prophet 13 years before hijrah (-13 H), when he was 40 years old (e.g. Al Husaini, 1996; Al Ghazaly, 1985). Concerning the date and the month, however, there is no agreement. According to Jabir and Ibnu Abbas mentioned above, that occurred on Monday 12 Rabiul Awal or 24 February 609 AD. While other opinion (e.g. Al Huasini, 1996) says that the descending of the Quran to make him being a prophet occurred on 17 Ramadhan. It is based on indication in Quran verse 8:41 that the Quran was revealed on "Furqan" day, the day when two armies met during Badar war on 17 Ramadhan. Another indication is mentioned in Quran verse 2:185 that the Quran was revealed on Ramadhan. If the

opinion of Jabir and Ibnu Abbas used for the day, 17 Ramadhan -13 H was Monday 25 August 609 AD.

Ash Shiddieqy (1977) mentioned that verses of *nubuawah* (declaring as a messenger) were revealed for the first time on Rabiul Awal, that is the first five verses of Al Alaq. Then the verses of *risalah* (declaring as a prophet) were revealed on 17 Ramadhan, that is the first verses of Al Muddathir. In history it is mentioned that when he received either the verses of *nubuawah* or the verses of *risalah*, the Prophet asked his wife Khadijah to cover him with a blanket. The fact that during summer in August the Prophet asking to be covered indicates that there was very severe feeling making him humanly frightened and trembled.

Concerning the *Isra' Mi'raj* when obligation of five times prayer received by the Prophet, there is no agreement when it happened. Majority of scholars follow the opinion of Ibnu Katsir adopted from weak hadith that *Isra' Mi'raj* took place on 27 Rajab -1 H (one year before Hijrah) (Al Husaini, 1996), that was Wednesday 15 October 620 AD. But, if the opinions of Jabir and Ibnu Abbas adopted that it took place on Monday 12 Rabi'ulawal, it means to be on 12 Rabiul Awal -3 H (3 years before Hijrah) or Monday 6 November 618 AD.

Hijrah of the Prophet PBUH took place on Rabi'ulawal year 13 *Bi'tsah* (13 years after declaring as a prophet) (Al Ghazaly, 1985). Leaving Mecca on 2 Rabiul Awal, he reached Madina on 12 Rabi'ulawal (Al Husaini, 1996). Reaching Madina 12 Rabi'ulawal 0 AH was on Monday 5 October 621, the same as opinion of Jabir and Ibnu Abbas that the day was Monday. Some authors of history books make confusion between hijrah and the first New Year after Hijrah. Haekal (1993) and Al Husaini (1996), for example, mention that hijrah took place on July. And Haekal says that the Prophet reached Madina on Friday. In fact, July was the New Year 1 Muharram 1 AH, Friday 16 July 622.

Fasting of Ramadhan began to be obligatory on Monday 2 Sha'ban 2 AH or 30 January 624 AD. It means the first fasting of Ramadhan was done on February-March. Using the program (by keeping in mind the probable error of 1 day), it is proved that the Prophet mostly did fasting 29 days as said in a hadith (see Table 2). One of the Eid ul Fitri was on Friday (i.e. on 15 March 625), that may be related to a hadith from Abu Hurairah saying that the Prophet said: " To day (Friday) there are shared two holidays, if you want, (eid prayer) is enough fulfilling your Friday prayer, but we do Friday prayer" (narrated by Abu Dawud).

Table 2 Ramadhan and Eid during the Prophet life

Year Hijri	Beginning of Ramadhan	Eid ul Fitri	Fasting day
2	Sunday, 26 Feb. 624	Monday, 26 Mar. 624	29
3	Thursday, 14 Feb. 625	Friday, 15 Mar. 625	29
4	Tuesday, 4 Feb. 626	Wednesday, 5 Mar. 626	29
5	Sunday, 25 Jan. 627	Monday, 23 Feb. 627	29
6	Thursday, 14 Jan. 628	Saturday, 13 Feb. 628	30
7	Monday, 2 Jan. 629	Wednesday, 1 Feb. 629	30
8	Friday, 22 Dec. 629	Sunday, 21 Jan. 630	30
9	Wednesday, 12 Dec. 630	Thursday, 10 Jan. 631	29
10	Sunday, 1 Dec. 631	Monday, 30 Dec. 631	29

History of Badar war was not consistent in the date and the day. Some scholars thought that Badar war happened on Friday 17 Ramadhan 2 AH (Al Ghazaly, 1985). In fact, 17 Ramadhan 2 H was on Tuesday 13 March 624 AD. Friday 17 Ramadhan was in year 1 AH, that was 25 March 623 AD. Confirming with other events, however, it was not possible that the war happened in year 1 AH. It seems the history of Badar war to be inaccurate in mentioning the day.

Uhud war took place on 15 Shawal 3 AH or Sunday 31 March 625 AD. In this war Muslims defeated after getting victory at first due to disobedience some groups of Muslim by leaving their strategic position. There were many victims in both sides. Then Abu Sufyan, before leaving war ground, challenge Muslims to meet again in Badar. The "little" Badar war took place in Sha'ban 4 AH (Al Ghazaly, 1985), that was January 626 AD, when there was shortage of food. The war failed due to Abu Sufyan was frightened and withdraw his army back to Mecca (Quran verse 3:172-174). It might be the hadith saying that the Prophet and his followers had just returned from little war going to big jihad, *jihadunnafs*, jihad attacking oneself desire during coming fasting Ramadhan, refers to this event. Similar to little Badar war, Khandaq war took place during winter and shortage of food on Shawal 5 AH or February 627 AD. The three weeks siege by *mushrikin* was ended after severe cold storm. While Tabuk war occurred during severe dry summer on Rajab 9 AH or October 630 AD. Quran verse 9:81 explain the hardship during this war in facing Roman.

The last verse of the Quran (Q 5:3) was revealed during wukuf in Arafah on 9 Dzulhijjah 10 AH that was Friday 6 March 632 AD. May be this event is adopted by some scholars to mention "Great Haj" in case the day of wukuf on Friday. Three months later, the Prophet passed away on 12 Rabi'ulawal 11 AH (Al Husaini, 1996). Calendar conversion program calculates that 12 Rabi'ulawal 11 AH must be Friday 6 June 632 AD. But many scholars, including Jabir and Ibnu Abbas, believed that the Prophet passed away on Monday. It must be 8 June 632 AD. The difference by two days cannot be explained just by probable error described earlier or *istikmal* (completing Safar to be 30 days). It might be there was general negligence due to sadness during sickness of the Prophet since Safar that made late in observing hilaal.

4. CONCLUDING REMARKS

Apart from general negligence, there is interesting thing concerning Monday 12 Rabiul Awal. Is that by chance or miracle that many important events happened on Monday 12 Rabi'ulawal? The consistence of the day and the date has proved that the Prophet was born, did hijrah, and passed away on that day and date. Although there is disagreement, it might be the first revelation of the Quran and Isra' Mi'raj also happened on 12 Rabiul Awal.

References

- Al Ghazaly, M. 1985, "Fiqhus Sirah" (Translated into Indonesian by Abu Laila & Muhammad Tohir), PT. Al Maarif, Bandung.
- Al Husaini, H. 1996, "Riwayat Kehidupan Nabi Besar Muhammad SAW" (History of the Life of Great Messenger Muhammad PBUH) (in Indonesian), Yayasan Al Hamidiy, Bandung
- Haekal, M. H. 1993, "Hayatu Muhammad" (Translated into Indonesian by Ali Audah), Litera Antarnusa, Jakarta.
- Ash Shiddieqy, H. 1977, "Tafsir Al Bayaan" Book 1 (in Indonesian), Al Maarif, Bandung